Pensieve header: Non-Commutative Gaussian Elimination, Day 2.

From http://www.math.toronto.edu/drorbn/classes/16-1750-ShamelessMathematica/About.html: **Possible Topics** (in no particular order). Whatever you may suggest, and the ~~Fibonacci numbers~~; ~~the Jones polynomial~~; ~~a more efficient Jones-algorithm~~; ~~a riddle on spheres~~; ~~Khovanov homology~~; Γ-calculus; the Hopf fibration; Hilbert's 13th problem; ~~non-commutative Gaussian elimination~~; free Lie algebras; the Baker-Campbell-Hausdorff formula; wacky numbers; ~~an order 4 torus~~; the Schwarz Lantern; knot colourings; the Temperley-Lieb pairing; the dodecahedral link; sound experiments; barycentric subdivisions; a Peano curve; braid closures and Vogel's algorithm; the insolubility of the quintic, phase portraits.

```
? Cycles
```

Cycles[{cyc_1, cyc_2, …}] represents a permutation with disjoint cycles cyc_i. ≫

```
n = 54;
g1 = Cycles[{{1, 18, 45, 28}, {2, 27, 44, 19},
     {3, 36, 43, 10}, {46, 52, 54, 48}, {47, 49, 53, 51}}];
g2 = Cycles[{{7, 16, 39, 30}, {8, 25, 38, 21}, {9, 34, 37, 12},
     {13, 15, 33, 31}, {14, 24, 32, 22}}];
g3 = Cycles[{{28, 31, 34, 48}, {29, 32, 35, 47}, {30, 33, 36, 46},
     {37, 39, 45, 43}, {38, 42, 44, 40}}];
g4 = Cycles[{{1, 3, 9, 7}, {2, 6, 8, 4}, {10, 54, 16, 13},
     {11, 53, 17, 14}, {12, 52, 18, 15}}];
g5 = Cycles[{{1, 13, 37, 46}, {4, 22, 40, 49}, {7, 31, 43, 52},
     {10, 12, 30, 28}, {11, 21, 29, 19}}];
g6 = Cycles[{{3, 48, 39, 15}, {6, 51, 42, 24}, {9, 54, 45, 33},
     {16, 18, 36, 34}, {17, 27, 35, 25}}];
```

```
? PermutationList
```

PermutationList[*perm*] returns a permutation list representation of permutation *perm*.

PermutationList[*perm*, *len*] returns a permutation list of length *len*. ≫

```
? PermutationCycles
```

PermutationCycles[*perm*] gives a disjoint cycle representation of permutation *perm*. ≫

```
? PermutationProduct
```

PermutationProduct[*a*, *b*, *c*] gives the product of permutations *a*, *b*, *c*. ≫

```
a_ ∘ b_ := PermutationProduct[a, b]
```

```
? InversePermutation
```

InversePermutation[*perm*] returns the inverse of permutation *perm*. ≫

```
? PermutationSupport
```

PermutationSupport[*perm*] returns the support of the permutation *perm*. ≫

```
? PermutationReplace
```

PermutationReplace[*expr*, *perm*] replaces each part in *expr* by its image under the permutation *perm*.

PermutationReplace[*expr*, *gr*] returns the list of images of *expr* under all elements of the permutation group *gr*. ≫

```
σ_ ∘ τ_ := PermutationProduct[τ, σ];
Feed[Cycles[{}]] := Null;
Feed[τ_] := Module[{i, j, k, l},
    i = Min[PermutationSupport[τ]];
    j = PermutationReplace[i, τ];
    If[Head[σ_{i,j}] === Cycles,
     Feed[InversePermutation[σ_{i,j}] ∘ τ],
     (*Else*) σ_{i,j} = τ;
     For[k = 1, k < n, ++k,
      For[l = k + 1, l ≤ n, ++l,
       If[Head[σ_{k,l}] === Cycles,
        Feed[σ_{i,j} ∘ σ_{k,l}]; Feed[σ_{k,l} ∘ σ_{i,j}]]
       ]]
    ]];
$RecursionLimit = ∞;

Table[Feed[g_α];
  ∏_{i=1}^{n} (1 + Count[Range[n], j_ /; Head[σ_{i,j}] == Cycles]), {α, 6}] // Timing
```

```
{14.8438, {4, 16, 159 993 501 696 000, 21 119 142 223 872 000,
  43 252 003 274 489 856 000, 43 252 003 274 489 856 000}}
```

```
ArrayPlot[
  Table[
   Which[
    Head[σ_{i,j}] === Cycles ∨ i == j, Red, i < j, Yellow, i > j, White],
   {j, n}, {i, n}],
  Mesh → True, Frame → True,
  FrameTicks → Range@n, AspectRatio → 9/16, ImageSize → 960
 ] // Rasterize
```



Math Questions.

1. Why are there so many yellow cells in column 1?
2. Why are there so many yellow cells in column 2?
3. Why are there “new” yellow cells in column 3?
4. Why are there “new” yellow cells in column 4?
5. Why is column 5 entirely yellow?
6. Why’s the big yellow gap between colulmn 10 and 18?

Mathematica Challanges.

1. Can you make the porgram doubly efficient?
2. Can you keep track of the actual “tricks”?
3. Can you find the true lengths of the “tricks”?
4. Can you fix these sad results?
5. Can you actually solve cubes?